

# Wave Functions: A wave-packet

Somehow, in this transition from classical mechanics (CM) to quantum mechanics (QM), we have lost our friends position and momentum ( $x$  and  $p$ ). Here I will show that we haven't really lost them, they are just a little less certain in QM.

## The wave function $\Psi(x)$

We start with a wave function with three parts: an amplitude  $A$ , a Gaussian envelope  $e^{-\frac{(x-x_0)^2}{2d}}$ , and a phase oscillation  $e^{ik_0(x-x_0)}$ . The envelope will serve as the position of our particle; we know it is somewhere around  $x_0$ , with an uncertainty of  $d$ . The oscillation with wave number  $k_0$  serves as our momentum, since de Broglie tells us that  $p = \hbar k$ .

In[1]:=

```

$$\Psi_u = A \text{Exp}[-(x - x_0)^2 / (2 d^2)] \text{Exp}[I k_0 (x - x_0)]$$

```

Out[1]=

```

$$A e^{i k_0 (x - x_0) - \frac{(x - x_0)^2}{2 d^2}}$$

```

(To keep *Mathematica* happy, we need to make explicit some assumptions about our parameters...)

In[2]:=

```

$$\$Assumptions = \{d > 0, \text{Element}[\{x, x_0, k_0, d\}, \text{Reals}]\}$$

```

Out[2]=

```

$$\{d > 0, (x | x_0 | k_0 | d) \in \text{Reals}\}$$

```

Before we get much further, let's normalize the wave function. By normalizing  $\Psi(x)$  we ensure that the probability of finding our particle *somewhere* is equal to 1.

In[3]:=

```

$$\begin{aligned} Px &= \text{Abs}[\Psi_u]^2; \\ \text{Anorm} &= \text{Solve}[\text{Integrate}[Px, \{x, -\text{Infinity}, \text{Infinity}\}] == 1, A] \end{aligned}$$

```

Out[4]=

```

$$\left\{ \left\{ A \rightarrow -\frac{1}{\sqrt{d} \pi^{1/4}} \right\}, \left\{ A \rightarrow \frac{1}{\sqrt{d} \pi^{1/4}} \right\} \right\}$$

```

We find that there are 2 possible values of  $A$ , one positive and one negative. There is no way to say which is correct, since in the end we can only measure  $|\Psi(x)|^2$  which means we only care about  $A^2$ . Let's update  $\Psi(x)$  with the positive value of  $A$ .

In[5]:=

```

$$\Psi = \Psi_u / . \text{Anorm}[[2]]$$

```

Out[5]=

```

$$\frac{e^{i k_0 (x - x_0) - \frac{(x - x_0)^2}{2 d^2}}}{\sqrt{d} \pi^{1/4}}$$


```

So that we can see what we are doing, here is a plot of the real part and magnitude of  $\Psi(x)$ . To move the particle around, change  $x_0$ . To make a well-localized particle with low average momentum, make  $d$  small (e.g.,  $\log D = -1$ ) and  $k_0$  close to 0. Making  $d$  large (e.g.,  $\log D = 2$ ) moving  $k_0$  away from zero makes a plane-wave-like wave function.

In[6]:=

```
Manipulate[Plot[{Re[Ψ], Abs[Ψ]} /. {A → 1, d → 10^logD, x0 → x0, k0 → k0},
  {x, -10, 10}, PlotRange → {-1, 1}, Filling → Axis, PerformanceGoal → Quality],
  {{logD, 0}, -1, 2, 0.1, Appearance → "Labeled"},
  {{x0, 2}, -10, 10, 0.1, Appearance → "Labeled"},
  {{k0, 4}, -10, 10, 0.1, Appearance → "Labeled"}]
```

Out[6]=



### The Fourier Transform of $\Psi(x)$ , $\tilde{\Psi}(k)$

The position-space wave function  $\Psi(x)$ , shown above, gives us a good idea of the uncertainty in  $x$ , but not a lot of information about the uncertainty in  $p$ . To get this, let's take the Fourier transform of  $\Psi(x)$  to find  $\tilde{\Psi}(k)$ .

In[7]:=

```
 $\tilde{\Psi} = 1 / \text{Sqrt}[2 \pi] \text{Integrate}[\Psi \text{Exp}[-I k x], \{x, -\text{Infinity}, \text{Infinity}\}]$ 
```

Out[7]=

$$\frac{\sqrt{d} e^{-\frac{1}{2} d^2 (k-k_0)^2 - i k x_0}}{\pi^{1/4}}$$

This momentum-space wave function  $\tilde{\Psi}(k)$ , can be separated in a way similar to  $\Psi(x)$ ... amplitude, envelope, phase.

In[8]:=

$$\tilde{\Psi} == \frac{\sqrt{d}}{\pi^{1/4}} \times e^{-\frac{1}{2} d^2 (k-k_0)^2} \times e^{-i k x_0}$$

Out[8]=

True


And it is properly normalized! (Has anyone heard of Parseval's Theorem?)

In[9]:= `Integrate[Abs[ $\tilde{\Psi}$ ]^2, {k, -Infinity, Infinity}]`

Out[9]= 1

In[10]:= `Manipulate[Plot[{Re[ $\tilde{\Psi}$ ], Abs[ $\tilde{\Psi}$ ]} /. {d → 10^logD, x0 → x0, k0 → k0},  
 {k, -10, 10}, PlotRange → {-1, 1}, Filling → Axis, PerformanceGoal → Quality],  
 {{logD, 0}, -1, 2, 0.1, Appearance → "Labeled"},  
 {{x0, 2}, -10, 10, 0.1, Appearance → "Labeled"},  
 {{k0, 4}, -10, 10, 0.1, Appearance → "Labeled"}]`

Out[10]=



You can see that the equation for  $\tilde{\Psi}(k)$  is essentially identical to  $\Psi(x)$  (only an overall phase of  $e^{i k_0 x_0}$  is missing). In particular, the uncertainty in  $k$  is now clear; the width of the Gaussian in frequency space is inversely proportional to  $d$ , so a narrow peak in  $\Psi(x)$  leads to a wide peak in  $\tilde{\Psi}(k)$ ! Putting this all together we see that  $\Delta x = d/\sqrt{2}$  and  $\Delta k = 1/\sqrt{2} d$  such that  $\Delta x \Delta k = 1/2$ , which de Broglie tells us means  $\Delta x \Delta p = \hbar/2$ . **Implication: the Uncertainty Principle need not be an axiom. It is simply the result of the mixed wave-particle nature of reality.**

### The Inverse Fourier Transform of $\tilde{\Psi}(k)$

Just to show that it works, let's take the inverse Fourier transform of  $\tilde{\Psi}(k)$  to get back to  $\Psi(x)$ ...

In[11]:=

$$\Psi_2 = 1 / \text{Sqrt}[2 \pi] \text{Integrate}[\tilde{\Psi} \text{Exp}[I k x], \{k, -\text{Infinity}, \text{Infinity}\}]$$

Out[11]=

$$\frac{e^{\frac{(x-x_0) \left( 2 i d^2 k_0 - x + x_0 \right)}{2 d^2}}}{\sqrt{d} \pi^{1/4}}$$

In[12]:=

$$\text{Simplify}[\Psi_2 == \Psi]$$

Out[12]=

True

MIT OpenCourseWare  
<http://ocw.mit.edu>

## 8.04 Quantum Physics I

Spring 2013

For information about citing these materials or our Terms of Use, visit: <http://ocw.mit.edu/terms>.